
'-----------------------------------------------------------------------------------------
' Name  : WS2811_V1.bas
' Datum: 28.11.2012
'
' Autor  : Stefan Kaldewey
' Email   : stefan.kaldewey@kt-elektronic.de
'-----------------------------------------------------------------------------------------

$regfile = "m16def.dat"                                     ' specify the used micro
$crystal = 10000000                                         ' used crystal frequency
$hwstack = 64                                               ' default use 32 for the hardware stack
$swstack = 32                                               ' default use 10 for the SW stack
$framesize = 32                                             ' default use 40 for the frame space

Declare Sub Send_frame_800(byval Number_of_ic As Byte)
Declare Sub Send_frame_400(byval Number_of_ic As Byte)

' enthält die PWM Werte für jeden Kanal
' die Größe des Arrays errechnet sich aus der Anzahl der WS2811 Controller * 3
' im Beispielprogramm 3 * 3 = 9
Dim Led_buffer(9) As Byte

' Hilfsvariablen zum Testen
Dim Rnd_value As Byte
Dim Loop_counter As Byte

' Konfiguration der Datenleitung

Sdi Alias Portd.6

Config Portd.6 = Output

' ---------  Hauptprogramm --------------
Do

     For Loop_counter = 1 To 9

        Led_buffer(loop_counter) = Rnd(30)

         'Call Send_frame_400(3)
         Call Send_frame_800(3)

     Next Loop_counter

     'Test mit einem Controller
     'Led_buffer(1) = 30     ' rot
     'Led_buffer(2) = 0       ' grün
     'Led_buffer(3) = 0       ' blau
     'Call Send_frame(1)

     Waitms 300

Loop
'--------- Ende Hauptprogramm --------------

Sub Send_frame_800(byval Number_of_ic As Byte)

   Local I As Byte
   Local J As Byte
   Local Last_channel As Byte
   Local Temp As Byte
   Local Mask As Byte

   Last_channel = Number_of_ic * 3
   Mask = 128

   Sdi = 0
   Waitms 1

   ' Bit Zeiten
   ' für eine 1, T1H 0.6 µs T1L 0.65 µs
   ' für eine 0, T0H 0.25 µs T0L 1.0 µs

      push r0                                               ' enthält das aktuelle Byte des Kanals
      push r16                                              ' Bitzähler
      push r17                                              ' Kanalzähler

      Loadadr Last_channel , X
      ld r17,x
      Loadadr Led_buffer(1) , X



Channel_loop_800:

            ld r0,x+
            ldi r16,8

Bit_loop_800:

            lsl r0
            brcc  BIT_0_800
Bit_1_800:
            Sbi $12 , 6
            nop
            nop
            nop
            nop
            cbi $12 , 6
            rjmp Next_bit_800

Bit_0_800:

            Sbi $12 , 6
            nop
            cbi $12 , 6

Next_bit_800:

            dec r16
            cpi r16,0
            brne  BIT_LOOP_800
            dec r17
            cpi r17,0
            brne  CHANNEL_LOOP_800

            pop r17
            pop r16
            pop r0

   Waitus 60

End Sub

'******************************************
'
'
'******************************************

Sub Send_frame_400(byval Number_of_ic As Byte)

   Local I As Byte
   Local J As Byte
   Local Last_channel As Byte
   Local Temp As Byte
   Local Mask As Byte

   Last_channel = Number_of_ic * 3
   Mask = 128

   Sdi = 0
   Waitms 1

   ' Bit Zeiten
   ' für eine 1, T1H 1.2 µs T1L 1.3 µs
   ' für eine 0, T0H 0.5 µs T0L 2.0 µs

      push r0                                               ' enthält das aktuelle Byte des Kanals
      push r16                                              ' Bitzähler
      push r17                                              ' Kanalzähler

      Loadadr Last_channel , X
      ld r17,x
      Loadadr Led_buffer(1) , X

Channel_loop_400:

            ld r0,x+
            ldi r16,8

Bit_loop_400:

            lsl r0
            brcc  BIT_0_400
Bit_1_400:



            Sbi $12 , 6
            nop
            nop
            nop
            nop

            nop
            nop
            nop
            nop
            nop
            nop

            cbi $12 , 6
            nop
            nop

            rjmp Next_bit_400
Bit_0_400:
            Sbi $12 , 6
            nop
            nop
            nop

            cbi $12 , 6
            nop
            nop
            nop

            nop
            nop
            nop
            nop
            nop
            nop

Next_bit_400:

            dec r16
            cpi r16,0
            brne  BIT_LOOP_400
            dec r17
            cpi r17,0
            brne  CHANNEL_LOOP_400

            pop r17
            pop r16
            pop r0

   Waitus 60

End Sub


